


C7. Tècniques de programació

Nom : _____

Heu de crear una carpeta en l'espai del servidor que teniu reservat amb el nom '**examen_vb**' i crear-hi a dins dues carpetes amb els noms '**ex_vb71**' i '**ex_vb72**'. Cadascuna de les carpetes ha de contenir els fitxers necessaris per a la correcta execució de l'aplicació, és a dir els fitxers del formulari principal '***.frm**' i el del projecte '***.vbp**'.

Recordeu que per obtenir la màxima puntuació cal que:

- Els dissenys que realitzeu s'ajustin a les especificacions i visualitzacions que es proposen
- Tingueu en compte els criteris de posar tabulacions i comentaris al codi font per fer els exercicis el màxim d'entenedors
- Les denominacions triades pels objectes han de correspondre amb la normativa

Exercici 1 [5 p]

Fixa't amb la imatge superior formada per dos frame's, el de l'esquerra conté un array de 64 textbox numerats del 0 al 63 i posat en 8 files i 8 columnes enganxa'ts els uns amb els altres.

El de la dreta conté dos botons d'opció excloents un de l'altre: *un sol color* i *aleatòris*, el primer permet escollir amb els botons de comandament un color (vermell, verd o blau) que al escollir-lo, pintarà cada una de les text box de l'esquerra amb una tonalitat del color escollit.


El segon botó (aleatòris) pintarà indefinidament cada un del text box de l'esquerra amb un color aleatòri.

A l'inici l'opció per defecte és *un sol color*.

L'aspecte de l'aplicació ha de ser com el de la figura adjunta. Trobareu a la unitat U: un fitxer executable anomenat **graella de colors.exe** que és un executable d'exemple.

Anomeneu el formulari amb el nom: '**principal.frm**' i el projecte amb: '**ex_vb71.vbp**' dins de la carpeta: '**ex_vb71**'

Exercici 2 [5 p]



Heu de fer una aplicació que sigui una calculadora amb les opcions de la figura superior.

Per les funcions trigonomètriques teniu les funcions Sin (), Cos () i Tan (), podeu trobar informació del seu funcionament en l'ajuda del visual. (Recordeu 2π radians = 360 °).

L'aspecte de l'aplicació ha de ser com el de la figura adjunta. Trobareu a la unitat U: un fitxer executable anomenat ***calculadora científica.exe*** que és un executable d'exemple.

L'aspecte de l'aplicació ha de ser com el de la figura adjunta.

Anomeneu el formulari amb el nom: '***principal.frm***' i el projecte amb: '***ex_vb72.vbp***' dins de la carpeta: '***ex_vb72***'

Solució als exercicis

Exercici 1

```
Option Explicit
Dim i, Color As Integer
Private Sub cmdVermell_Click()
    frmGraella.Caption = "Gairebé tots els vermells"
    Color = 0
    For i = 0 To 63
        txtGraella(i).BackColor = RGB(Color, 0, 0)
        Color = Color + 4
    Next i
End Sub
Private Sub cmdBlau_Click()
    frmGraella.Caption = "Gairebé tots els blaus"
    Color = 0
    For i = 0 To 63
        txtGraella(i).BackColor = RGB(0, 0, Color)
        Color = Color + 4
    Next i
End Sub
Private Sub cmdSortir_Click()
    End
End Sub
Private Sub cmdVerd_Click()
    frmGraella.Caption = "Gairebé tots els verds"
    Color = 0
    For i = 0 To 63
        txtGraella(i).BackColor = RGB(0, Color, 0)
        Color = Color + 4
    Next i
End Sub
Private Sub Form_Load()
    Randomize
    optUn.Value = True
    Timer1.Enabled = False
    frmGraella.Caption = "Tria un color?"
End Sub
Private Sub optAleatori_Click()
    cmdVermell.Enabled = False
    cmdVerd.Enabled = False
    cmdBlau.Enabled = False
    Timer1.Enabled = True
    frmGraella.Caption = "Colors aleatòris"
End Sub
Private Sub optUn_Click()
    cmdVermell.Enabled = True
    cmdVerd.Enabled = True
    cmdBlau.Enabled = True
    Timer1.Enabled = False
    frmGraella.Caption = "Tria un color?"
End Sub
Private Sub Timer1_Timer()
    Dim Color1, Color2, Color3, Pos As Integer
    Pos = Int(Rnd * 63)
    Color1 = Int(Rnd * 255)
    Color2 = Int(Rnd * 255)
    Color3 = Int(Rnd * 255)
    txtGraella(Pos).BackColor = RGB(Color1, Color2, Color3)
End Sub
```

Exercici 2

```
Option Explicit
Dim num1, num2, resultat As Double
Dim inici, opera As Boolean
Dim pi As Double
Private Sub cmdEsborra_Click()
    txtDisplay.Text = "0"
    num1 = 0
    num2 = 0
End Sub
Private Sub cmdNumero_Click(Index As Integer)
If inici Then
    txtDisplay.Text = ""
End If
inici = False
txtDisplay.Text = txtDisplay.Text + cmdNumero(Index).Caption
End Sub
Private Sub cmdOperacio_Click(Index As Integer)
Dim i As Integer
    inici = True
    pi = 3.14159265358979
    If Index = 6 Then
        num1 = Val(txtDisplay.Text)
        opera = True
    End If
    Select Case Index
        Case 0
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(Sin(num1 * 2 * pi / 360))
        Case 1
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(Cos(num1 * 2 * pi / 360))
        Case 2
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(Tan(num1 * 2 * pi / 360))
        Case 3
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(1 / num1)
        Case 4
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(num1 * num1)
        Case 5
            num1 = Val(txtDisplay.Text)
            txtDisplay.Text = Str(num1 * num1 * num1)
        Case 7
            num1 = Val(txtDisplay.Text)
            If num1 > 150 Or (num1 < 1 And num1 > 0) Then
                MsgBox "No puc fer el factorial d'aquest número", vbCritical,
"Avis"
            Else:
                resultat = 1
                For i = num1 To 1 Step -1
                    resultat = resultat * i
                Next i
                txtDisplay.Text = Str(resultat)
            End If
        End Select
    If Index = 8 And opera = True Then
        num2 = Val(txtDisplay.Text)
        resultat = 1
        inici = True
        opera = False
        For i = 1 To num2
```

```
        resultat = resultat * num1
    Next i
    txtDisplay.Text = Str(resultat)
End If
End Sub
Private Sub Command1_Click()
    End
End Sub

Private Sub Form_Load()
    inici = True
    opera = True
    txtDisplay.Text = "0"
End Sub
```